
Abstract
This is a paper that introduces the storage and
retrieval of graphs in The SAS System. The
audience will be shown how to store graphs in a
catalog, how to retrieve the graph and print using
PROC GREPLAY, and how to format the stored
graph using templates. Interactive and non-
interactive use of PROC GREPLAY will be
covered.

Storing Graphs

Graphs created in SAS/GRAPH will be stored in
a SAS catalog. A SAS catalog is a SAS file that
is used to store SAS specific objects, such as
graphs, frames, macros, and formats.
1. To create a SAS catalog containing a

graph created in SAS/GRAPH, simply
include the option
GOUT=<libname>.<catname> in your
PROC Gxxxx statement, where libname
is the library name and catname is the
catalog name. Please note the following:

2. If libname is not specified, then WORK is
assumed.

3. If the GOUT option is not specified, then
the graphics output is directed to the
catalog WORK.GSEG.

Optional: if you add the statement
NAME=’anything’ to your graph statement
options, where ’anything’ is a descriptive name
up to 8 characters long. This will place your plot
in the catalog with a descriptive name.

Please note: if the name already exists, the last
character receives a ‘1’.

Example:

* Declare library names in UNIX  
  environment.;
libname sampsio
        ’/sas/sas612/samples/graph’;
libname graphcat
        ’/sas/sas612/sasgraph’;

proc summary data=sampsio.giris2 nway;
  var petallen petalwid sepallen
      sepalwid;
  class spec_no;
  output out=giris2
  mean=petallen petalwid sepallen
       sepalwid;
run;

symbol interpol=join;

* Create two plots, put in the 
  GRAPHCAT.GIRIS2 catalog.;
* Give the plots names of PETALS and
  SEPALS.;
proc gplot data=giris2 uniform
     gout=graphcat.giris2;
  title ’Petal Length and Width’;
  plot petallen*spec_no
       / name=’petals’;
  plot2 petalwid*spec_no;
run;

  title ’Sepal Length and Width’;
  plot sepallen*spec_no
       / name=’sepals’;
  plot2 sepalwid*spec_no;
run; quit;

The catalog GRAPHCAT.GIRIS2 now contains
two graphs, named PETALS and SEPALS,
respectively.

Replaying

You now have graphs stored in SAS catalog.
Now what? PROC GREPLAY is the tool to use to
replay, reformat, and reuse these graphs. PROC
GREPLAY can be used interactively or
programmatically.



To replay interactively, submit the following code
to pull up the GREPLAY window:
proc greplay igout=graphcat.giris2;
run; quit;

The following GREPLAY window will now
appear:

To simply display each graph on a full screen,
put an ’s’ in the ’Sel’ column next to each graph’s
name, then press enter.

However, if you want to place the graphs into a
template, then follow these steps:
1. In the field ’TC’, you need to type the

name of the ’Template Catalog’. Try
typing: SASHELP.TEMPCAT

2. In the field ’Template’, type the name of
the template (found in the specified
template catalog) you want to use. Try
typing: V2S

3. Finally, under the heading ’Sel’, type ’1’
next to the ’PETALS’ graph, and ’2’ next
to the ’SEPALS’ graph. Press enter, and
watch the graph be replayed.

To run the same graph non-interactively, you
need to specify the ’TC’, the ’template’, and the
positions, as above. In addition, however, the
’NOFS’ option must be specified so that a
GREPLAY window does not appear.

Notice in the following code that the TREPLAY
line specifies the how the graphs are displayed
on a Template, whereas the REPLAY line will
simply replay a graph onto the full screen (similar
to using an ’s’ interactively).
proc greplay igout=graphcat.giris2
     tc=sashelp.templt
     template=v2s
     nofs;

  * Replay into TEMPLATE.;
  treplay 1:petals 2:sepals;

  * Replay into a whole screen.;
  replay petals;
  replay sepals;

run; quit;

Templates

A template catalog is simply a catalog that
contains templates that you can use. A catalog of
templates is supplied with The SAS System, and
is found in SASHELP.TEMPLT. This catalog
contains most of the common templates. To see
a list of the templates found in that (or any other)
catalog, and a sample of each, try running this
code:
proc greplay nofs tc=sashelp.templt;
  list tc;
  preview v3s;
run;
quit;

’List TC’ will print a list of the templates in the
specified template catalog to the SAS log.
’Preview v3s’ will create a preview of the
template named ’v3s’.

TIP #1: use ’Preview _all_’ to create a preview for
all templates in the given catalog.

TIP #2: If the template you desire isn’t available
in that SASHELP.TEMPLT catalog, you can
create a new template by using the TDEF line
command. See online help or the SAS/GRAPH
Software Reference Volume 2 (chapter 36) for
more details.



Deleting Old Graphs
You can use PROC GREPLAY also to
delete members of a catalog. Interactively,
on the main screen, simply put a ’d’ on the
’Sel’ line next to the graph name you want
to delete. Programmatically, use the
following code:
proc greplay igout=graphcat.giris2
     nofs;
  delete petals sepals;
quit;

Graphics Editor

You can customize your graphs by using the
SAS/GRAPH editor, and then save the results
into your catalog. Follow these steps:
1. In the ’GRAPH’ window, simply type

’EDIT’ on the command line.
2. Add annotations, lines, and other objects

using the toolbar provided.

3. 
3. Go to File...Save As...Graphic to save

into a SAS catalog.

What Next?

Here is an exercise you could try on your own to
become more familiar with PROC GREPLAY:
1. Create two graphs. Suggestions: Create

a vertical bar chart from
SAMPSIO.REJECTS containing the sum
of rejects for each location. Also, create
a pie chart from SAMPSIO.REJECTS
containing the sum of rejects for each
location.

2. Store those two graphs into a permanent
graphics catalog.

3. Replay them in a template that you think
looks nice.

4. Add some circles and text where you see
appropriate.

5. Try these on one or more operating
systems, if possible.

Reference
SAS/GRAPH Software, Reference, Volume 2,
Chapter 36 (pages 1191-1258).

